
The µBotino Microcontroller Board  

by Ro-Bot-X Designs 

 

Introduction.  

The µBotino Microcontroller Board is an Arduino compatible board for small robots. The 5x5cm (2x2”) 
size and the built in 3 pin connectors for sensors and servos, along with the 1 Amp dual H-bridge, 
makes the µBotino board easy to use for any small project. 

 

Features:  

• SN754410 dual H-bridge, 1 Amp per motor, uses digital pins 5-8 and Timer0 for PWM 
• 5 digital 3 pin connectors (GND, Vcc, Signal), digital pins 0-4 
• 5 servo 3 pin connectors (GND, Vxx, Signal), digital pins 9-13, uses power from jumper J1 
• 6 sensor 3 pin connectors (GND, Vcc, Signal), analog pins 0-5, uses regulated 5V 
• GND pin next to the UART pins for 3 pin serial cable 
• FTDI cable connector 
• ISP hardware programmer connector 
• power LED (green) 
• pin D13 LED (red) 
• jumper J1 to select the power to the servo pins 
• jumper J2 to disable motor power when programming 
• jumper J3 to disable pin D13 LED in case a servo is connected to the same pin 
• polarized power connector 
• On/Off switch 
• 1Amp 5V LDO voltage regulator 
• reset button 
 

 

 

 



Assembly.  

The kit is easy to assemble. You will need to following basic tools: 

• fine tip soldering iron, low temperature 
• diagonal cutters 
• rosin core solder 

 

The kit contains all the parts needed for the uBotino board: 

 

Start to solder the parts that have the smallest height, then the parts that a bit taller and so on. Here is 
how I did it: 

1. Insert the 0.1uF ceramic capacitors (104 marking) 
one by one in their marked place and bend the leads 
outwards to keep them from falling. 
 

 



2. Turn the board upside down, put it on the table and 
solder the leads. The solder joint should look like a 
shiny cone and cover the pad ring completely. Using 
the diagonal cutters cut the leads as close as possible 
to the solder joint. They should look like this: 
 

 
3. Insert the 28 and 16 pin sockets in their marked 
place, bend 2 opposite corner pins to keep them from 
falling, turn the board upside down and solder the 
other 2 opposite corner pins. Make sure the sockets 
are flush with the board. If they are, straighten up the 
bent pins and solder all pins. 
 

 
4. Insert the crystal in its marked place, turn the board 
upside down and solder. Trim the leads close to the 
solder joint. 
 

 
5. Insert the Reset button in its place, turn the board 
upside down and solder. You don`t need to cut the 
leads. 
 
 
Insert the On/Off switch in it's place and solder the 
leads. 
 

 



6. Insert the blue (22j marking) ceramic capacitors in 
their place next to the crystal, bend the leads so they 
don’t fall, turn the board and solder. Trim the leads 
close to the solder joint. 

 
7. Insert the Green LED with the flat side towards the 
top of the board, in the place marked Pwr LED. Insert 
the Red LED with the flat side towards the top of the 
board, in the place marked D13 LED. Turn the board 
upside down and solder. Trim the leads close to the 
solder joint. 

 
8. Locate the 2 1k ohm resistors (brown, black, red), 
bend one lead along the resistor and insert them in 
their marked place near the LEDs. Bend the leads so 
they don’t fall. Locate the 10k ohm resistor (brown, 
black, orange), bend one lead along the resistor and 
insert it in its marked place near the Reset button. 
Bend the leads so it doesn’t fall, turn the board upside 
down, place it on the table and solder the leads. Trim 
the leads close to the solder joints. 

 
9. Insert the Power switch in its marked place and 
solder it. Trim the leads. 
 

 



10. Insert 3 groups of 6 pin headers in the place 
marked for sensors. Use a female header to keep 
them straight and aligned. If you don’t have a female 
header handy, use the small conductive foam where 
the H-bridge was inserted. 

 
11. Turn the board upside down, place it on the table 
carefully so the pin leads stick out equally from the 
board. Solder one pin in each group, then inspect the 
headers so they look straight and aligned. If they are, 
solder all pins. If not, re-heat the soldered pins and 
align the headers properly. 
 

 
12. Repeat the procedure for the 3 groups of 5 pins 
needed for the servo connectors. 
 

 
13. Locate one group of 6 pins and 2 groups of 5 pins. 
Insert them in the place for digital sensors, keep them 
aligned with a female header or with the conductive 
foam, then solder. 
 

 



14. Locate 2 groups of 2 pins and insert them in the 
place marked for motors. Keep them aligned with the 
foam and solder, then repeat for the 3 pin jumper J1. 
 

 
15. Insert the last group of 6 pins in the place marked 
FTDI. Use the foam to keep them aligned and then 
solder them. 

 
16. Repeat the procedure for the 2 groups of 2 pins in 
their marked place for jumpers J2 and J3. 

 
17. Repeat the procedure for the 2 groups of 3 pins in 
their marked place for ISP connector. 

 



18. Find 3 electrolytic capacitors marked 10µF. Insert 
them in their marked place with the short lead in the 
hole marked with an – sign. Make sure they are 
completely inserted, then bend the leads. Turn the 
board upside down, place it on the table and solder 
the leads. Trim the leads as close to the solder joint as 
possible. 

 
19. Look for the 2 capacitors marked 220µF. Insert 
them in their marked place near the top of the board, 
with the short lead in the hole marked with the – sign. 
Bend the leads, solder and trim them. 

 
20. Insert the white power connector with the tab 
towards the Red LED and solder the leads. 

 
21. Insert the voltage regulator in its marked place at 
the top of the board, with the metallic tab towards the 
exterior of the board. Solder the leads then trim them. 

 



22. You’re almost done! The bottom of the board 
should look like this picture -> 
 
 
 
 
 
 
Check out your solder joints with a multi-meter, watch 
out for shorts between adjacent pins. If there is too 
much solder, use a solder wick to suck up the excess. 

 
23. Battery leads. My battery holder has similar 
connectors with a 9V battery, so I got a 9V connector 
with leads, crimped the 2 terminals to the leads and 
inserted the Red lead terminal in the hole marked with 
a 1 (or arrow) in the housing. The Black lead terminal 
goes right next to it. Verify the polarity with a 
multi-meter, Red should be Positive and Black should 
be Negative. 
 
Warning:  If you accidentally swap them, you may 
burn your H-bridge and the microcontroller. 
  
24. Make sure the On/Off switch lever is positioned 
towards the voltage regulator. That is the Off position. 
Insert the battery connector and make sure the Red 
wire is towards the Power switch.  
 
Flip the lever of the Power switch towards the battery 
connector (On position). The Green LED will light up 
and the Red LED will flicker, then it will start blinking 
once a second. 
 
 
You're done! Congratulations! Now build your robot! 

 
 

 

Usage.  

Install the µBotino board on your robot and connect the motors, sensors, servos and other used 
peripherals. Remove the J2 jumper so the robot does not start moving after programming. 

Use a FTDI Basic 5V board or a FTDI Cable to connect the µBotino to the computer. The FTDI will 
power the µBotino so turning the Power switch On is not necessary unless you are using a lot of power 
hungry sensors that need more current than the USB can provide.  

In the Arduino IDE, select the board as Arduino Duemilanove and the serial port to the one your FTDI 



uses. Load your sketch and press Upload. You will see the D13 LED flicker and then the FTDI board 
LEDs will flicker, but if you’re using an FTDI cable there will be no visible indicator that the sketch is 
loading. Look on the Arduino IDE screen for the message Uploading Done.  

Remove the FTDI board or cable, place the J2 jumper back and turn the Power switch On. The robot 
will start moving!  

 

Program sample. 

 

Here is a sample program used on my Compactino robot: 

// Compactino, a small robot with the uBotino board 
// 2 Pololu micro motors 100:1 and small wheels, 
// one HXT900 micro servo, one Sharp IR sensor, 
// 2 AAA battery holders 
// 
// Arduino pinout: 
// 
// Shield  Funct Arduino  ATmega168      Arduino Funct Shield 
//                       +-----\/----+ 
//          Reset       1| PC6   PC5 |28  D19 A5  SCL   
//          Rx    D0    2| PD0   PC4 |27  D18 A4  SDA   
//          Tx    D1    3| PD1   PC3 |26  D17 A3         
//          Int0  D2    4| PD2   PC2 |25  D16 A2         
//          Int1  D3    5| PD3   PC1 |24  D15 A1         
//                D4    6| PD4   PC0 |23  D14 A0        IR sensor 
//                      7| VCC   GND |22   
//                      8| GND  AREF |21   
//          Xtal        9| PB6  AVCC |20   
//          Xtal       10| PB7   PB5 |19  D13      SCK  LED       
//      M1A OC0B  D5   11| PD5   PB4 |18  D12      MISO Pan servo 
//      M2A OC0A  D6   12| PD6   PB3 |17  D11 OC2A MOSI        
//      M2B       D7   13| PD7   PB2 |16  D10 OC1B       
//      M1B       D8   14| PB0   PB1 |15  D 9 OC1A       
//                       +-----------+ 
// 
 
 
#include <Servo.h> 
 
//Inputs/outputs 
#define Motor_1_PWM   5 // digital pin 5    // Right Motor 
#define Motor_1_Dir   8 // digital pin 8 
#define Motor_2_PWM   6 // digital pin 6   // Left Motor 
#define Motor_2_Dir   7 // digital pin 7 
 
#define IR_Pin  14  // digital pin 14 (analog pin 0) 
#define PanPin  12 
#define LedPin  13 
 
#define SR 1 //Sharp Short Range sensor 
#define MR 2 //Sharp Medium Range sensor 
#define LR 3 //Sharp Long Range sensor 
 
#define center 90 
 



//Variables 
byte dir=0; 
byte speed1=250; 
byte speed2=255; 
int turn90=110; 
int turn45=55; 
int straight=500; 
int stopTime=200; 
int IRdistance=0; 
int treshold=20; //20cm min distance 
 
Servo Pan;  
 
//-------------------------------------------------------------------------- 
void setup() {  
   
  // set motor pins as output and LOW so the motors are braked 
  pinMode(Motor_1_PWM, OUTPUT); 
  pinMode(Motor_1_Dir, OUTPUT); 
  pinMode(Motor_2_PWM, OUTPUT); 
  pinMode(Motor_2_Dir, OUTPUT); 
  Stop(); 
 
  Pan.attach(PanPin); 
  Pan.write(center); //90 degrees, looking straight ahead 
  StepDelay(); 
 
  pinMode(LedPin, OUTPUT);       
  digitalWrite(LedPin, LOW);   
 
  Forward(); 
}  
 
void loop(){ 
  Drive(); 
} 
 
//-------------------------------------------------------------------------- 
void Drive(){ 
  IRdistance=Read_Sharp_Sensor(MR, IR_Pin); 
  Serial.print("IRdistance "); 
  Serial.println(IRdistance); 
  if (IRdistance<10){ 
    Stop(); 
    StepDelay(); 
    TurnAround(); 
  } 
  if (IRdistance<treshold){ 
    Stop(); 
    StepDelay(); 
    Avoid(); 
    Forward(); 
  } 
  delay(50); 
} 
 
void TurnAround(){ 
    Reverse(); 
    Pan.write(center); 
    StepDelay(); 
    Stop(); 
    Left(); 
    delay(turn90); 



    delay(turn90); 
    Stop(); 
    StepDelay(); 
    Forward(); 
} 
 
void Avoid(){ 
  int prev=0; 
  dir=2; 
  for (byte i=0; i<5; i++){ 
    Pan.write(i*45); //turn 45 degrees at a time, from 0 to 180 degrees 
    StepDelay(); 
    IRdistance=Read_Sharp_Sensor(MR, IR_Pin); 
    if (IRdistance>prev){ 
      dir=i; 
      prev=IRdistance; 
    } 
  } 
  Pan.write(center); 
  StepDelay(); 
  switch (dir){ 
    case 0: 
      Right(); 
      delay(turn90); 
      Stop(); 
      StepDelay(); 
      break; 
    case 1: 
      Right(); 
      delay(turn45); 
      Stop(); 
      StepDelay(); 
      break; 
    case 2: 
      Forward(); 
      break; 
    case 3: 
      Left(); 
      delay(turn45); 
      Stop(); 
      StepDelay(); 
      break; 
    case 4: 
      Left(); 
      delay(turn90); 
      Stop(); 
      StepDelay(); 
      break; 
  } 
  delay(500); 
}   
 
// Read Sensors 
int Read_Sharp_Sensor(byte model, byte pin) { 
  int value = 0; 
  value = analogRead(pin); 
  switch (model) { 
    case SR: //short range, aka GP2D120 (4-30cm) 
      return (2914/(value+5))-1;  
      break; 
    case MR: //medium range, aka GP2D12 (10-80cm) 
      return 5*1384.4*pow(value,-.9988); //multiplied by 5, different sensor 
      break; 



    case LR: //long range, aka GP2Y0A02YK (20-150cm) 
      return 11441*pow(value,-.9792);  
      break; 
  } 
} 
 
void StepDelay() { 
    for (byte t=0; t<10; t++){ 
      delay(20); 
    } 
} 
 
//++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++ 
void Forward(){ 
  digitalWrite(Motor_1_Dir, LOW); // forward 
  digitalWrite(Motor_2_Dir, LOW); // forward 
  analogWrite(Motor_1_PWM, speed1); //  
  analogWrite(Motor_2_PWM, speed2); // 
  return; 
} 
 
void Reverse(){ 
  digitalWrite(Motor_1_Dir, HIGH); // reverse 
  digitalWrite(Motor_2_Dir, HIGH); // reverse 
  analogWrite(Motor_1_PWM, 255-speed1); //  
  analogWrite(Motor_2_PWM, 255-speed2); // 
  return; 
} 
 
void Right(){ 
  digitalWrite(Motor_1_Dir, HIGH); // reverse 
  digitalWrite(Motor_2_Dir, LOW); // forward 
  analogWrite(Motor_1_PWM, 255-speed1); //  
  analogWrite(Motor_2_PWM, speed2); // 
  return; 
} 
 
void Left(){ 
  digitalWrite(Motor_1_Dir, LOW); // forward 
  digitalWrite(Motor_2_Dir, HIGH); // reverse 
  analogWrite(Motor_1_PWM, speed1); //  
  analogWrite(Motor_2_PWM, 255-speed2); // 
  return; 
} 
 
void Stop() 
{ 
  digitalWrite(Motor_1_PWM, LOW); 
  digitalWrite(Motor_1_Dir, LOW); 
  digitalWrite(Motor_2_PWM, LOW); 
  digitalWrite(Motor_2_Dir, LOW); 
  return; 
}   


